

## A Short Program for Teaching Chemical Equilibrium

Juan Miguel Campanario

*Departamento de Física, Universidad de Alcala  
28871 Alcala de Henares, Madrid, Spain*

Reyes Ballesteros

*Instituto de Ciencias de la Educacion  
Universidad Autonoma de Madrid  
28043 Cantoblanco, Madrid, Spain*

### Abstract

A short GW/BASIC program for the automatic calculation of equilibrium concentrations is presented. This program can be used to solve problems with an arbitrary number of reactants and products. The program uses an iterative algorithm. Common and sometimes hard to understand simplifications are avoided. Once the calculation have been carried out, the equilibrium concentrations of reactants and products are shown and a Kc is computed from these concentrations.

Teaching chemical equilibrium takes up a large part of the chemistry curriculum. The problems and questions related to chemical equilibrium can be presented in various ways, often involving ionic species in solution. One of the main objectives of chemical education is that students understand chemical equilibrium as a dynamic equilibrium in which two semi-reactions, direct and inverse, take place. It is also desirable that students learn to use the mass action law and Le Chatelier's principle to predict the result of the possible variations of variables (pressure, volume,...) involved in chemical equilibria. Another main objective is that intermediate level students learn how to compute equilibrium concentrations from initial concentrations of chemical species involved in a given reaction. To do so students are generally taught to write the Kc formula with the dissociation rate,  $\alpha$ , as an unknown. Typical problems are carefully chosen to obtain easily-solved first grade or, at the most, second grade equations. The numerical values of Kc and initial concentrations are also chosen in such a way that, if necessary, and its powers are small and can be ignored with regard to initial concentrations. Students sometimes do not understand properly when this can and cannot be done. All of the above result from the students using rote instead of meaningful learning strategies for calculation. Thus many students have misconceptions about chemical equilibrium (Wheeler and Kass, 1978), even though they are able to solve the numeric problem properly. The situation is even worse: these misconceptions can be identified also in some preservice science teachers (Furiò and Ortiz, 1983).


When there are more than four or five chemical species involved, or when simplifications cannot be carried out, or the stoichiometric coefficients are greater than 2, chemical equilibrium problems are hard to solve with the above methods. Due to these difficulties, these kinds of problems are rarely proposed in the chemistry class or laboratory. Thus it is not strange that many students believe that the "real" problems are one thing and the problems that the chemistry teacher gives in the classroom are very different.

Some methods for the automatic calculation of equilibrium concentrations have been proposed, and some FORTRAN program listings have been published (see Bard and King, 1965 and Feenstra, 1979). These programs address the many simultaneous equilibria problems and use hard matrix methods to solve a set of many simultaneous equations. Consequently, they are of little use in teaching at intermediate levels since, with these approaches, students use the programs like a "magic wand" that solves the problems by magic.

A new approach using computational methods is presented in this article. The short BASIC program described here is based on a more accurate version of method used by Devore (1988). In the Devore method there are some terms containing powers greater than 1 that are dropped. These simplifications are avoided here because, although they aid in calculation, they decrease the logical understanding and meaning of the subject to be studied.

### Derivation of a General Formula

Consider a chemical reaction in which  $N$  reactants react to give  $M$  products according to the following equation:



When equilibrium is reached, the constant  $K_c$  is obtained by using the concentrations of all chemical species in the reaction by means of the following well-known expression:

$$K_c = \frac{\prod_{i=1}^M (C_{2i})_{eq}^{a_{2i}}}{\prod_{i=1}^N (C_{1i})_{eq}^{a_{1i}}}$$

Consider the reaction of some chemical species ( $A_{1i}$  y  $A_{2i}$ ), with initial concentrations of  $C_{1i}$  y  $C_{2i}$ . Depending on the values of these initial concentrations, three situations are found: (1) The chemical species on the left side (reactants),  $A_{1i}$ , react and transform into the chemical species on the right,  $A_{2i}$ , until equilibrium is reached; (2) the chemical species on the right side (products),  $A_{2i}$ , react and transform into the chemical species on the left,  $A_{1i}$ ; and (3) the initial concentrations,  $C_{1i}$  and  $C_{2i}$ , may be such that the system is already in equilibrium.

If we define  $K(x)$  as:

$$K(x) = \frac{\prod_{i=1}^M (C_{2i} + a_{2i} \cdot x)^{a_{2i}}}{\prod_{i=1}^N (C_{1i} - a_{1i} \cdot x)^{a_{1i}}} \quad (1)$$

a function  $F(x)$  can be stated as follows:

$$F(x) = K_c - K(x)$$

The products  $a_{1i} \cdot x$  and  $a_{2i} \cdot x$  in equation 1 represent the number to be added or subtracted from each initial concentration to reach the equilibrium constant,  $K_c$ . Thus the interpretation of  $x$  is simple: it represents the mol/l of a given chemical species that is formed or transformed divided by the respective stoichiometric coefficient. If the initial concentrations,  $C_{1i}$  and  $C_{2i}$ , are such that the system is in situation (1) above,  $x$  will be greater than zero. If the system is in situations (2) or (3), the  $x$  variable will be negative or zero respectively. Now the problem is to find an  $x$ -value such that the value of  $F(x)$  is zero. This can be achieved by means of a method of calculation. The short BASIC program presented here uses the bisection method. Once the root of  $F(x)$  has been obtained, the equilibrium concentrations can be quickly computed. In principle, the two extreme values that limit the range of  $x$  are the lowest values of the initial concentrations divided by the respective stoichiometric coefficient on each side of the chemical reaction (minimum of  $C_{1i}/a_{1i}$  and minimum of  $C_{2i}/a_{2i}$  respectively). Thus, if the absolute value of a product  $a_{1i} \cdot x$  or  $a_{2i} \cdot x$  were greater than the respective  $C_{1i}$  or  $C_{2i}$  and the sign was opposite, negative concentrations would be obtained in equation 1, and this does not make sense.

The calculation of a value for  $x$  that solves the function  $F(x)$  starts by replacing the initial concentrations in equation 1, when  $x$  is zero. If the equilibrium concentrations are equal to the initial concentrations, the first value of  $x$  is the root of  $F(x)$  and further calculations are not necessary. In general,  $F(x)$  will be positive or negative and it will be necessary to find out the interval of values of  $x$  to be explored. When  $x = 0$ , the function  $F(x)$  is positive and so, to make  $F(x)$  zero, the definitive value of  $x$  must both be positive and greater than zero and less than the minimum of  $C_{1i}/a_{1i}$  (left side of the reaction). If  $F(x)$  is negative,  $x$  must be negative and (in absolute value) greater than 0 and less than the minimum of  $C_{2i}/a_{2i}$  (right side of the reaction). Once the interval of possible  $x$ -values have been determined, the bisection method is used. The interval between the extreme values is halved and those values in which the product of the function values at the interval limits is always negative are taken as the new extremes. This process is repeated until the absolute value of the rate between  $K(x)$  and  $K_c$  differs by 1 within a given small amount of tolerance, say plus or minus  $10^{-5}$ . At this moment the "calculated" constant will be nearly equal to the  $K_c$  constant and the problem will have been solved. Another possible criterium for convergence can be taken—given small value of tolerance (say plus or minus  $10^{-6}$ ) above which  $F(x)$  is not considered as zero. This criterion would be the same for all the equilibrium problems. But when the  $K_c$  constant was too small this criterion would not work well. In this case the

difference between  $K(x)$  and  $K_c$  could be lesser than the tolerance, which will not obviate the error of  $K(x)/K_c$  being very high.

### Running the BASIC Program

We give the BASIC program for equilibrium calculation in the Appendix. The computational segment of the program is very compact and occupies only 12 lines. This program can be easily modified for adaptation to other systems or languages. The program uses double precision to obtain more accurate results. If an equilibrium constant is very strong (i.e. the dissociation constant of HCl), the program assumes an arbitrarily high value (i.e.  $10^{10}$ ) for the corresponding  $K_c$ . The program checks the user input to make sure that concentrations, stoichiometric coefficients and equilibrium constant are appropriate. Once the calculation have been carried out, the equilibrium concentrations of reactants and products are shown and a  $K_c$  is computed from these concentrations. In Figure 1 a screen dump of a calculation is shown.

```
ENTER THE NUMBER OF CHEMICAL SPECIES ON THE LEFT SIDE OF
  THE EQUATION
(REACTANTS) ? 2
ENTER THE NUMBER OF CHEMICAL SPECIES ON THE RIGHT SIDE OF
  THE EQUATION
(PRODUCTS) ? 3
ENTER THE EQUILIBRIUM CONSTANT, Kc. (I.E.: .00012, 0.00012, 1.2e-4,
  12e-5,...) ? 1.23e-5
IF Kc IS VERY HIGH THEN HIT <H>

ENTER THE INITIAL CONCENTRATION (Mol/l) OF REACTANT NUMBER 1
(I.E.: .00012, 0.00012, 1.2e-4, 12e-5,...)
? .4

ENTER THE STOICHIOMETRIC COEFFICIENT OF REACTANT NUMBER 1
? 2

ENTER THE INITIAL CONCENTRATION (Mol/l) OF REACTANT NUMBER 2
(I.E.: .00012, 0.00012, 1.2e-4, 12e-5,...)
? .24

ENTER THE STOICHIOMETRIC COEFFICIENT OF REACTANT NUMBER 2
? 3
```

**Figure 1.** Screen dump of the calculation of equilibrium concentrations for equation  $2A + 3B = 2C + D + 4E$  ( $K_c = 1.23 \cdot 10^{-5}$ )

(Figure 1 continued)

ENTER THE INITIAL CONCENTRATION (Mol/l) OF PRODUCT NUMBER 1  
(I.E.: .00012, 0.00012, 1.2e-4, 12e-5,...)

? 0

ENTER THE STOICHIOMETRIC COEFFICIENT OF PRODUCT NUMBER 1

? 1

ENTER THE INITIAL CONCENTRATION (Mol/l) OF PRODUCT NUMBER 2  
(I.E.: .00012, 0.00012, 1.2e-4, 12e-5,...)

? 1e-3

ENTER THE STOICHIOMETRIC COEFFICIENT OF PRODUCT NUMBER 2

? 3

ENTER THE INITIAL CONCENTRATION (Mol/l) OF PRODUCT NUMBER 3  
(I.E.: .00012, 0.00012, 1.2e-4, 12e-5,...)

? .18

ENTER THE STOICHIOMETRIC COEFFICIENT OF PRODUCT NUMBER 3

? 4

#### EQUILIBRIUM CONCENTRATIONS

REACTANT NUMBER 1 CONCENTRATION (Mol/l) = .3648194122314453

REACTANT NUMBER 2 CONCENTRATION (Mol/l) = .187229118347168

PRODUCT NUMBER 1 CONCENTRATION (Mol/l) =

1.759029388427734D-02

PRODUCT NUMBER 2 CONCENTRATION (Mol/l) =

5.377088165283203D-02

PRODUCT NUMBER 3 CONCENTRATION (Mol/l) = .2503611755371094

Kc FROM EQUILIBRIUM CONCENTRATIONS = 1.230004122179591D-05

ACTUAL Kc = .0000123

**Appendix**  
Basic Program for Equilibrium Calculations

```

10 '-----
20 'CHEMICAL EQUILIBRIUM CALCULATIONS
30 'IBM/PC OR COMPATIBLE (GW/BASIC)
40 '-----
50 CLS: KEY OFF
60 DIM C#(2,50),A#(2,50),N(2),A$(2), M#(2),P#(2),MI(2),S#(2)
70 A$(1)="REACTANT": A$(2)="PRODUCT ": S#(1)=-1: S#(2)=1
80 '-----
90 'C#: Initial concentrations
100 'A#: Stoichiometric coefficients
110 'N : Chemical species on each side of the equation
120 'M#: Lowest value of concentration
130 'MI: Index of chemical species whose concentration is lowest
140 '-----
150 LOCATE 8,5: PRINT "ENTER THE NUMBER OF CHEMICAL SPECIES ON THE
    LEFT SIDE OF THE EQUATION": LOCATE 9,5: PRINT "(REACTANTS) "": INPUT
    N(1)
160 LOCATE 10,5: PRINT "ENTER THE NUMBER OF CHEMICAL SPECIES ON THE
    RIGHT SIDE OF THE EQUATION": LOCATE 11,5: PRINT "(PRODUCTS) "":
    INPUT N(2)
170 IF N(1)<1 OR N(2)<1 THEN E$="THE NUMBER OF REACTANTS OR
    PRODUCTS MUST BE EQUAL OR GREATER THAN 1": GOSUB 620: GOTO 150
180 LOCATE 12,5: PRINT "ENTER THE EQUILIBRIUM CONSTANT, Kc.": CHR$(13);
    " (I.E.: .00012, 0.00012, 1.2e-4, 12e-5,...)": CHR$(13); " IF Kc IS VERY HIGH
    THEN HIT <H>"
190 LOCATE 15,5: INPUT KC$
200 IF KC$="H" OR KC$="h" THEN KC#= 1D+10 ELSE KC#=VAL(KC$)
210 IF KC#<1D-38 THEN E$="Kc IS TOO SMALL" : GOSUB 620: GOTO 180
220 FOR J=1 TO 2
230 M#(J)=1D+38: MI(J)=0
240 FOR I=1 TO N(J)
250 CLS
260 LOCATE 6,5: PRINT "ENTER THE INITIAL CONCENTRATION (Mol/l) OF ";
    A$(J);" NUMBER"; I; CHR$(13); SPACES$(4);"(I.E.: .00012, 0.00012, 1.2e-4, 12e-
    5,...)"
270 LOCATE 8,5: INPUT C#(J,I)
280 IF C#(J,I)<0 THEN E$="THE INITIAL CONCENTRATION (Mol/l) OF
    "+A$(J)+" NUMBER" +STR$(I)+" IS WRONG": GOSUB 620: GOTO 260
290 LOCATE 11,5: PRINT "ENTER THE STOICHIOMETRIC COEFFICIENT OF ";
    A$(J);" NUMBER "": I;CHR$(13)
300 LOCATE 12,5: INPUT A#(J,I): IF A#(J,I)<1 OR A#(J,I)>>INT(A#(J,I)) THEN
    E$="THE STOICHIOMETRIC COEFFICIENT OF "+A$(J)+" NUMBER "
    +STR$(I)+" IS WRONG": GOSUB 620: GOTO 290
310 IF C#(J,I)<M#(J) THEN M#(J)=C#(J,I): MI(J)=I
320 NEXT I

```

```

330 NEXT J
340 IF M#(1) < 1D-38 AND M#(2) < 1D-38 THEN E$="THE INITIAL
      CONCENTRATIONS OF A REACTANT AND OF A PRODUCT ARE TOO
      SMALL": GOSUB 620: GOTO 220
350 ML# = -M#(2)/A#(2,MI(2)): MR# = M#(1)/A#(1,MI(1))
360 CLS: LOCATE 11,33: PRINT "I AM COMPUTING"
370 M# = 0
380 FOR J=1 TO 2
390 P#(J) = 1
400 FOR I=1 TO N(J)
410 P#(J) = P#(J)*(C#(J,I) + M#*A#(J,I) *S#(J))^A#(J,I)
420 NEXT I
430 NEXT J
440 IF P#(1) > 1D-38 THEN K# = P#(2)/P#(1) ELSE MR# = M#: GOTO 480
450 ER# = ((K#/KC#)-1)
460 IF ABS(ER#) < 1D-5 THEN GOTO 490
470 IF KC# < K# THEN MR# = M# ELSE ML# = M#
480 M# = (MR# + ML#)/2: GOTO 380
490 CLS
500 PRINT SPACE$(4);"EQUILIBRIUM CONCENTRATIONS";CHR$(13);CHR$(13)
510 FOR J=1 TO 2
520 FOR I=1 TO N(J)
530 PRINT SPACE$(4);A$(J);" NUMBER "; I;" CONCENTRATION (Mol/l) = ";
      (C#(J,I) + M#*A#(J,I)*S#(J))
540 NEXT I
550 PRINT CHR$(13)
560 NEXT J
570 PRINT CHR$(13);SPACE$(4);"Kc FROM EQUILIBRIUM CONCENTRATIONS =
      " ; K#;CHR$(13);SPACE$(4);"ACTUAL Kc"; SPACE$(26);" = "; KC#
580 PRINT CHR$(13);SPACE$(4);"ANOTHER PROBLEM? (Y/N)"
590 B$=INKEY$: IF B$="" THEN GOTO 590
600 IF B$="Y" OR B$="y" THEN CLS: GOTO 150
610 IF B$="n" OR B$="N" THEN STOP ELSE GOTO 590
620 CLS
630 BEEP
640 LOCATE 10,1: PRINT E$
650 LOCATE 13,1: PRINT "HIT <SPACE>"
660 B$=INKEY$: IF B$ <> CHR$(32) THEN GOTO 660
670 CLS: RETURN

```

### References

- Bard, A.J., and King, D.M. (1965). General digital computer program for chemical equilibrium calculations. *Journal of Chemical Education* 42 (3) : 127-131.
- Devore, J.A. (1988). A simple numerical method for solving complex equilibria. *Journal of Chemical Education* 65 (10) : 868-870.
- Feenstra, T.P. (1979). A note on the calculation of concentrations in the case of many simultaneous equilibria. *Journal of Chemical Education* 56 (2) : 104-105.

- Furio, C. J. and Ortiz, E. (1983). Persistencia de errores conceptuales en el estudio del equilibrio químico. *Revista de Enseñanza de las Ciencias* 1 : 15-20.
- Wheeler A.E. and Kass H. (1978). Students' misconceptions in chemical equilibrium. *Science Education* 62 (2) : 223-232.